

## Lessons learned

### Running head: Working with dates in Excel


By Lucas Timmons

If you've ever had to deal with dates while using Excel you know what a pain they can be. Here are a few operations you can use to make them easier to use in your data analysis.

Before we start, it's important to know how Excel deals with dates. For our purposes, we'll use the 1900 date system. (Note: this article will deal only with dates from 1900-01-01 onward. There are many strategies to mitigate the lack of support from Excel for earlier dates however they will no be discussed here.) The system uses serial numbers for each date. 1 is Jan. 1, 1900. 2 is Jan. 2, 1900, and so on. So for example 42,518 would be May 28, 2016.

To find a serial number, you can use the date function. In an empty cell type **=DATE(2016,05,28)**. Your cell should display 42,518. You've serialized a date using date function. The first argument passed to the function is the year, the second argument is the month, and the third is the day. The output is the serial number.

Because the dates are stored this way, it is easy to do math to calculate the difference between dates. However, working with date serial numbers isn't very practical.



To work with dates more easily, we'll format them into something more familiar. Highlight the column your serial number dates are in, and then bring up the "Format Cells" dialogue and click on the "Number" tab. In the category list chose "date", and then pick the format that works best for you. We'll be using yyyy-mm-dd. The image below shows the conversion.

Serial Number	Date			Serial Number	Date
1	1			1	1900-01-01
2	2			2	1900-01-02
3	3			3	1900-01-03
42450	42450			42450	2016-03-21

Now that you've set the cells to recognize your input as dates, you can just type in the date in the yyyy-mm-dd format, and not have to worry about doing any conversion at all.

### Add or subtract days from a date

Because Excel stores dates as serial numbers, adding or subtracting days is easy. Using “+” sign or the “-“ sign in a standard cell formula, Excel will calculate the new date and show it in the format that you've selected. (Row 3 is for demonstrative purposes only, to show the formulas used)

	A	B	C
1	Today	90 Days from now	90 days ago
2	2016-03-21	2016-06-19	2015-12-22
3	FORMULAS	=A2+90	=A2-90

### Add or subtract months from a date

Again, Excel storing dates as serial numbers helps here. Instead of just using a “+” or “-“ formula to add or subtract months, use EDATE. The first argument you pass to the formula is the start date, the second

argument is how many months you want to add or subtract. Excel will calculate the new date, and then show it in the format that you've selected. (Row 3 is for demonstrative purposes only, to show the formulas used)

	A	B	C
1	Today	3 months from now	3 months ago
2	2016-03-21	2016-06-21	2015-12-21
3	FORMULAS	=EDATE(A2,3)	=EDATE(A2,-3)

### Add or subtract a combination of days, months, and years to a date

Adding or subtracting years to a date requires no formula, just common integer math. If you want to add a combination of days, weeks and years it can get confusing, fast. That's where using the DATE formula, along with the addition or subtraction arguments becomes useful.

	A	B	C
1	Today	One year, two months & three days from now	One year, two months & three days ago
2	2016-03-21	2017-05-24	2015-01-18
3	FORMULAS	=DATE(YEAR(A2)+1,MONTH(A2)+2,DAY(A2)+3)	=DATE(YEAR(A2)-1,MONTH(A2)-2,DAY(A2)-3)

In this formula **=DATE(YEAR(A2)+x,MONTH(A2)+y,DAY(A2)+z)**  
 The letters "x", "y", "z" represent integers for years, months and days you want to add or subtract. The result of the formula is your new date. You can use bot addition and subtraction in the same formula:

	A	B
1	Today	+ two years, - six months & + eight days
2	2016-03-21	2017-09-29
3	FORMULAS	=DATE(YEAR(A2)+2,MONTH(A2)+6,DAY(A2)+8)

Now that we have a basic understanding of how to find dates, we will look at how to do calculations with them.

### Finding days between dates

Excel offers a few functions for doing calculations with dates.

The first is the **=DAYS(end\_date, start\_date)** function. This is different from the **=DAY()** function above - don't forget the "s". Use the function with the end date as the first argument, and the start date as the second argument, and it will return the number of days between the two dates. Make sure this output is in a cell stored as a number, and not as a date! If the cell is stored as a date it will take your result and assume it is a date serial number and display it as a date.

If you would rather have the result listed in years, months and days rather than just the total days, use the DATEDIF function.

**=DATEDIF(start\_date, end\_date, "interval")**. The start date must always occur before the end date. There are six possible values for the interval: "d", "m", "y", "ym", "yd" and "md".

"d" - Days between dates

"m" - Months between dates

"y" - Years between dates

"ym" - Months between dates, ignoring year (July 2015 to August 2016 = 1 not 13)

"yd" - Days between dates ignoring year (January 1, 2015 to January 10, 2016 = 9)

"md" - Days between dates ignoring month (January 1, 2015 to February 10, 2015 = 9)

	A	B	C	D	E	F	G	H
1	Date 1	Date 2	DAYS BETWEEN DATES					
2	2016-01-01	2017-11-27		696	<-- NOT STORED AS A DATE!			
3	FORMULAS			=DAYS(B2,A2)				
4								
5	Date 1	Date 2	MONTHS BETWEEN DATES					
6	2016-01-01	2017-11-27		22	<-- NOT STORED AS A DATE!			
7	FORMULAS			=DATEDIF(A6,B6,"m")				
8								
9	Date 1	Date 2	YEARS BETWEEN DATES					
10	2016-01-01	2017-11-27		1	<-- NOT STORED AS A DATE!			
11	FORMULAS			=DATEDIF(A10,B10,"y")				
12								
13	Date 1	Date 2	MONTHS BETWEEN DATES (IGNORING YEAR)					
14	2016-01-01	2017-11-27		10	<-- NOT STORED AS A DATE!			
15	FORMULAS			=DATEDIF(A14,B14,"ym")				
16								
17	Date 1	Date 2	DAYS BETWEEN DATES (IGNORNING YEAR)					
18	2016-01-01	2017-11-27		26	<-- NOT STORED AS A DATE!			
19	FORMULAS			=DATEDIF(A18,B18,"md")				
20								
21	Date 1	Date 2	COMBINED					
22	2016-01-01	2017-11-27	1 years, 10 months, 26 days	<-- NOT STORED AS A DATE!				
23	FORMULAS		=DATEDIF(A22,B22,"y") & " years, " & DATEDIF(A22,B22,"ym") & " months, " & DATEDIF(A22,B22,"md") & " days"					

Chaining the DATEDIF formulas together in one cell using the ampersand to concatenate the outputs with strings as seen in row 23 will give you the most readable result.

## Work days

Now that you can find the number of days between dates, you might want only include weekdays. Luckily Excel has some functions that can help with that, too.

The first is =**NETWORKDAYS**(start\_date, end\_date, [holidays]). It returns the number of days in a period, minus the weekends and holidays. This example ignores the holidays, and uses a Monday to

Friday work week. The function still works if you omit the argument in brackets.

	A	B	C	D	E
1	Date 1	Date 2	WORK DAYS BETWEEN DATES		
2	2016-01-01	2017-11-27		497	<-- NOT STORED AS A DATE!
3	FORMULAS		=NETWORKDAYS(A2,B2)		
4					

The second is =NETWORKDAYS.INTL(start\_date, end\_date, [weekend], [holidays]). It returns the number of work days in a period, minus the custom weekend and holidays.

	A	B	C	D	E
1	Date 1	Date 2	WORK DAYS BETWEEN DATES		
2	2016-01-01	2017-11-27		496	<-- NOT STORED AS A DATE!
3	FORMULAS		=NETWORKDAYS.INTL(A2,B2,1,C5:C7)		
4					
5		Christmas	2016-12-25		
6		New Years	2017-01-01		
7		Canada Day	2016-07-01		
8					

The holidays can be defined from a range of dates. In this example the range is c5:c7. Notice how this example's result is 496, and the previous example's result is 497. The difference is Canada Day. Christmas and New Years are on weekends in this example, but Canada day is not. It is removed as a work day.

The important difference between =NETWORKDAYS() and =NETWORKDAYS.INTL() is being able to define a custom weekend.

	A	B	C	D	E
1	Date 1	Date 2	WORK DAYS BETWEEN DATES		
2	2016-01-01	2017-11-27		594	<-- NOT STORED AS A DATE!
3	FORMULAS		=NETWORKDAYS.INTL(A2,B2,17,C5:C7)		
4					
5		Christmas		2016-12-25	
6		New Years		2017-01-01	
7		Canada Day		2016-07-01	

Compare this example with the previous one. The formulas are almost identical. The only difference is the weekend argument. In the first example, it is set to 1 and in the second it is set to 17. There is a major difference in the total number of work days as you can see in the C2 cells.

The weekend argument uses a built-in list of days. The table below outlines what each argument represents. In our examples the first uses a standard Saturday and Sunday weekend (1) and the second uses a Saturday only weekend (17).



Weekend number	Weekend days
1 or omitted	Saturday, Sunday
2	Sunday, Monday
3	Monday, Tuesday
4	Tuesday, Wednesday
5	Wednesday, Thursday
6	Thursday, Friday
7	Friday, Saturday
11	Sunday only
12	Monday only
13	Tuesday only
14	Wednesday only
15	Thursday only
16	Friday only
17	Saturday only

The final functions we'll look at are **=WORKDAY(start\_date, days, [holidays])** and **=WORKDAY.INTL(start\_date, days, [weekend], [holidays])**. These functions will tell you how many work days there are in the next “x” number of days. Essentially, they work the same as the NETWORKDAYS functions, but instead of an end date, you specify a number of days in the future from the start date. The output gives you the date the days are completed.

	A	B	C
1	Date 1	Days	DATE A NUMBER OF WORK DAYS FROM A DATE
2	2016-01-01	300	2017-02-27
3	FORMULAS		=WORKDAY(A6,B6,C10:C12)
4			
5			
6		Christmas	2016-12-25
7		New Years	2017-01-01
8		Canada Day	2016-07-01

## Conclusion

These are some basic date manipulation techniques that you can use when exploring data. Keep in mind that working with dates in Excel can be frustrating. But now, armed with this knowledge, hopefully you have an easier time of it.

[Lucas Timmons](#) is a Toronto-based reporter with The Canadian Press. You can reach him at: [lucas@lucastimmons.com](mailto:lucas@lucastimmons.com)